Project 1

Spartan6 - DSP48A1



Submitted by:

Muhammed Atef Mokhtar

Under the supervision of Eng. Kareem Waseem

Contents

Introduction

This report presents the complete design, verification, and implementation flow of the **DSP48A1** slice, a critical hardware block within the Xilinx Spartan-6 FPGA family designed to accelerate math-intensive applications.

The primary objective of this project is to develop a highly configurable Register Transfer Level (RTL) model of the DSP48A1 block using Verilog. The model accurately captures the internal architecture, including the pre-adder/subtracter, the 18x18 multiplier, the 48-bit post-adder/accumulator, and the programmable pipeline registers.

Following the RTL design phase, a comprehensive functional verification is performed using **QuestaSim**. A directed testbench is developed to validate the design across four distinct data paths, dynamically controlled by the 8-bit OPMODE signal. The simulation process is fully automated using a TCL DO file to ensure efficiency and reproducibility.

Finally, the project covers the physical implementation flow using **Xilinx Vivado**. Due to the high number of I/O ports required by the DSP48A1, the design is targeted towards the xc7a200tffg1156-3 FPGA part. A timing constraint is applied to ensure the design operates successfully at a target clock frequency of 100 MHz. This report systematically documents the successful outcomes of the RTL simulation, schematic elaboration, synthesis, implementation, and code linting, confirming a robust and clean digital design.

1-RTL Code

Top Module

```
module DSP48A1 (A,B,D,C,CLK,CARRYIN,OPMODE,BCIN,RSTA,RSTB,RSTM,
RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE,CEA,CEB,CEM,CEP,CEC,CED,
CECARRYIN,CEOPMODE,PCIN,BCOUT,PCOUT,P,M,CARRYOUT,CARRYOUTF);

parameter A0REG       = 0;
parameter A1REG       = 1;
parameter B0REG       = 0;
parameter B1REG       = 1;
parameter CREG        = 1;
parameter DREG        = 1;
parameter PREG        = 1;
parameter MREG        = 1;
parameter CARRYINREG  = 1;
parameter CARRYOUTREG = 1;
parameter OPMODEREG   = 1;
parameter CARRYINSEL  = "OPMODE5"; // Carry-in source select
parameter B_INPUT     = "DIRECT";  // B port source select
parameter RSTTYPE     = "SYNC";    // Reset type for all regs
// Data Ports
input  [17:0] A, B, D;
input  [47:0] C;
output [35:0] M;
output [47:0] P;
input         CARRYIN;
output        CARRYOUT, CARRYOUTF;
// Control Input Ports
input        CLK;
input  [7:0] OPMODE;
// Clock Enable Input Ports
input CEA, CEB, CEC, CECARRYIN, CED, CEM, CEOPMODE, CEP;
// Reset Input Ports
input RSTA, RSTB, RSTC, RSTCARRYIN, RSTD, RSTM, RSTOPMODE, RSTP;
// Cascade Ports
input  [17:0] BCIN;
output [17:0] BCOUT;
input  [47:0] PCIN;
output [47:0] PCOUT;

// Stage 1
```

```verilog
// D path
wire [17:0] d_pipe_out;
BLK #(.WIDTH(18),.RSTTYPE(RSTTYPE)) D_STAGE
    (.D(D),.SEL(DREG),.CLK(CLK),.RST(RSTD),.CE(CED),.BLK_OUT(d_pipe_out));
// B0 path
wire [17:0] b_input_mux;
assign b_input_mux = (B_INPUT == "DIRECT")  ? B    :
                     (B_INPUT == "CASCADE") ? BCIN : 18'd0;
wire [17:0] b0_pipe_out;
BLK #(.WIDTH(18),.RSTTYPE(RSTTYPE)) B0_STAGE
    (.D(b_input_mux),.SEL(B0REG),.CLK(CLK),.RST(RSTB),.CE(CEB),.BLK_OUT(b0_pipe_out));
// A0 path
wire [17:0] a0_pipe_out;
BLK #(.WIDTH(18),.RSTTYPE(RSTTYPE)) A0_STAGE
    (.D(A),.SEL(A0REG),.CLK(CLK),.RST(RSTA),.CE(CEA),.BLK_OUT(a0_pipe_out));
// C path
wire [47:0] c_pipe_out;
BLK #(.WIDTH(48),.RSTTYPE(RSTTYPE)) C_STAGE
    (.D(C),.SEL(CREG),.CLK(CLK),.RST(RSTC),.CE(CEC),.BLK_OUT(c_pipe_out));

// Stage 2
// OPMODE path
wire [7:0] opmode_pipe_out;
BLK #(.WIDTH(8),.RSTTYPE(RSTTYPE)) OPMODE_STAGE
    (.D(OPMODE),.SEL(OPMODEREG),.CLK(CLK),.RST(RSTOPMODE),.CE(CEOPMODE),.BLK_OUT(opmode_pipe_o
ut));
// Pre-add/sub
wire [17:0] pre_adder_out;
assign pre_adder_out = (opmode_pipe_out[6]) ? (d_pipe_out - b0_pipe_out)
                                            : (d_pipe_out + b0_pipe_out);
wire [17:0] pre_mux_out;
assign pre_mux_out = (opmode_pipe_out[4]) ? pre_adder_out : b0_pipe_out;

// Stage 3
// B1 path
wire [17:0] b1_pipe_out;
BLK #(.WIDTH(18),.RSTTYPE(RSTTYPE)) B1_STAGE
    (.D(pre_mux_out),.SEL(B1REG),.CLK(CLK),.RST(RSTB),.CE(CEB),.BLK_OUT(b1_pipe_out));
// A1 path
wire [17:0] a1_pipe_out;
BLK #(.WIDTH(18),.RSTTYPE(RSTTYPE)) A1_STAGE
    (.D(a0_pipe_out),.SEL(A1REG),.CLK(CLK),.RST(RSTA),.CE(CEA),.BLK_OUT(a1_pipe_out));
```

```
// Stage 4
// B1 output
assign BCOUT = b1_pipe_out;
//Multiplier output
wire [35:0] mult_raw_out;
assign mult_raw_out = b1_pipe_out * a1_pipe_out;
// M path
wire [35:0] m_pipe_out;
BLK #(.WIDTH(36),.RSTTYPE(RSTTYPE)) M_STAGE
    (.D(mult_raw_out),.SEL(MREG),.CLK(CLK),.RST(RSTM),.CE(CEM),.BLK_OUT(m_pipe_out));
// M FPGA output
assign M = m_pipe_out;
// Carry-in source mux
wire carry_src_mux;
assign carry_src_mux = (CARRYINSEL == "OPMODE5") ? opmode_pipe_out[5] :
                       (CARRYINSEL == "CARRYIN") ? CARRYIN            : 1'b0;
// CYI
wire cin_reg_out;
BLK #(.WIDTH(1),.RSTTYPE(RSTTYPE)) CYI_STAGE
    (.D(carry_src_mux),.SEL(CARRYINREG),.CLK(CLK),.RST(RSTCARRYIN),.CE(CECARRYIN),.BLK_OUT(cin
_reg_out));

// Stage 5
wire [47:0] dab_concat;
assign dab_concat = {D[11:0], A[17:0], B[17:0]};
// X mux
wire [47:0] x_mux_out;
assign x_mux_out = (opmode_pipe_out[1:0] == 2'd0) ? 48'd0      :
                   (opmode_pipe_out[1:0] == 2'd1) ? {12'd0, m_pipe_out} :
                   (opmode_pipe_out[1:0] == 2'd2) ? P           : dab_concat;
// Z mux
wire [47:0] z_mux_out;
assign z_mux_out = (opmode_pipe_out[3:2] == 2'd0) ? 48'd0    :
                   (opmode_pipe_out[3:2] == 2'd1) ? PCIN      :
                   (opmode_pipe_out[3:2] == 2'd2) ? P         : c_pipe_out;
// Post-adder/subtracter
wire [47:0] post_adder_sum;
wire        post_adder_cout;
assign {post_adder_cout, post_adder_sum} =
    (opmode_pipe_out[7]) ? (z_mux_out - (x_mux_out + cin_reg_out))
                         : (z_mux_out + x_mux_out  + cin_reg_out);
// CYO
BLK #(.WIDTH(1),.RSTTYPE(RSTTYPE)) CYO_STAGE
```

```
    (.D(post_adder_cout),.SEL(CARRYOUTREG),.CLK(CLK),.RST(RSTCARRYIN),.CE(CECARRYIN),.BLK_OUT(
CARRYOUT));
assign CARRYOUTF = CARRYOUT;
// P path
BLK #(.WIDTH(48),.RSTTYPE(RSTTYPE)) P_STAGE
    (.D(post_adder_sum),.SEL(PREG),.CLK(CLK),.RST(RSTP),.CE(CEP),.BLK_OUT(P));
assign PCOUT = P;
endmodule
```

Building Block

```
module BLK (D,SEL,CLK,RST,CE,BLK_OUT);
parameter WIDTH = 18;
parameter RSTTYPE = "SYNC";
input SEL,CLK,RST,CE;
input [WIDTH-1:0] D;
reg [WIDTH-1:0] Q;
output [WIDTH-1:0] BLK_OUT;
generate
    if (RSTTYPE == "SYNC") begin
        always @(posedge CLK) begin
            if (RST) begin
                Q<=0;
            end
            else if (CE) begin
                Q<=D;
            end
        end
    end
    else if (RSTTYPE == "ASYNC") begin
        always @(posedge CLK or posedge RST) begin
            if (RST) begin
                Q<=0;
            end
            else if (CE) begin
                Q<=D;
            end
        end
    end
endgenerate
assign BLK_OUT = (SEL)? Q : D ;
endmodule
```

2-Testbench Code

```
module DSP48A1_tb();
parameter A0REG = 0 ;
parameter A1REG = 1 ;
parameter B0REG = 0 ;
parameter B1REG = 1 ;
parameter CREG = 1;
parameter DREG = 1 ;
parameter MREG = 1 ;
parameter PREG = 1 ;
parameter CARRYINREG = 1 ;
parameter CARRYOUTREG = 1 ;
parameter OPMODEREG = 1;
parameter CARRYINSEL = "OPMODE5" ;
parameter B_INPUT = "DIRECT" ;
parameter RSTTYPE = "SYNC" ;

reg [17:0] A, B, D;
reg [47:0] C, PCIN;
reg CLK, CARRYIN;
reg [7:0] OPMODE;
reg RSTA, RSTB, RSTM, RSTP, RSTC, RSTD, RSTCARRYIN, RSTOPMODE;
reg CEA, CEB, CEM, CEP, CEC, CED, CECARRYIN, CEOPMODE;
reg [17:0] BCIN;
wire [17:0] BCOUT;
wire [47:0] PCOUT, P;
wire [35:0] M;
wire CARRYOUT, CARRYOUTF;

// Variables for Path 3
reg [47:0] prev_P;
reg prev_CARRYOUT;

DSP48A1
#(.A0REG(A0REG),.A1REG(A1REG),.B0REG(B0REG),.B1REG(B1REG),.CREG(CREG),.DREG(DREG),.MREG(MREG),
.PREG(PREG),.CARRYINREG(CARRYINREG),.CARRYOUTREG(CARRYOUTREG),
        .OPMODEREG(OPMODEREG),.CARRYINSEL(CARRYINSEL),.B_INPUT(B_INPUT),.RSTTYPE(RSTTYPE)) DUT
(.A(A),.B(B),.D(D),.C(C),.CLK(CLK),.CARRYIN(CARRYIN),.OPMODE(OPMODE),
        .BCIN(BCIN),.RSTA(RSTA),.RSTB(RSTB),.RSTM(RSTM),.RSTP(RSTP),.RSTC(RSTC),.RSTD(RSTD),.R
STCARRYIN(RSTCARRYIN),.RSTOPMODE(RSTOPMODE),.CEA(CEA),.CEB(CEB),.CEM(CEM),
        .CEP(CEP),.CEC(CEC),.CED(CED),.CECARRYIN(CECARRYIN),.CEOPMODE(CEOPMODE),.PCIN(PCIN),.B
COUT(BCOUT),.PCOUT(PCOUT),.P(P),.M(M),.CARRYOUT(CARRYOUT),.CARRYOUTF(CARRYOUTF));
```

```
initial begin
    CLK = 0;
    forever
    #1 CLK = ~CLK;
end

initial begin
    // Initialize and reset signals
    RSTA = 1; RSTB = 1; RSTM = 1; RSTP = 1; RSTC = 1; RSTD = 1; RSTCARRYIN = 1; RSTOPMODE = 1;
    CEA = 1; CEB = 1; CEM = 1; CEP = 1; CEC = 1; CED = 1; CECARRYIN = 1; CEOPMODE = 1;
    A = 18'hABCD; B = 18'h1234; C = 48'hDEADBEEF; D = 18'h5678; CARRYIN = 1; BCIN = 18'hFFF;
PCIN = 48'h123456;
    OPMODE = 8'hFF;

    @(negedge CLK);

    // Reset Self-checking
    if (P !== 0 || M !== 0 || CARRYOUT !== 0 || BCOUT !== 0)
        $display("FAIL [Reset]: Outputs are not zero");
    else
        $display("PASS [Reset]: Outputs are zero");

    // Release resets
    RSTA = 0; RSTB = 0; RSTM = 0; RSTP = 0; RSTC = 0; RSTD = 0; RSTCARRYIN = 0; RSTOPMODE = 0;

    // Test case 1: Verify DSP Path 1
    A = 20; B = 10; C = 350; D = 25; CARRYIN = 1; BCIN = 18'hABC; PCIN = 48'h5A5A5A;
    OPMODE = 8'b11011101;
    repeat(4) @(negedge CLK);
    if (BCOUT !== 18'hf || M !== 36'h12c || P !== 48'h32 || PCOUT !== 48'h32 || CARRYOUT !==
1'b0)
        $display("FAIL [Path 1]");
    else
        $display("PASS [Path 1]");

    // Test case 2: Verify DSP Path 2
    A = 20; B = 10; C = 350; D = 25; CARRYIN = 1; BCIN = 18'h333; PCIN = 48'hABCDEF;
    OPMODE = 8'b00010000;
    repeat(3) @(negedge CLK);
    if (BCOUT !== 18'h23 || M !== 36'h2bc || P !== 48'h0 || PCOUT !== 48'h0 || CARRYOUT !==
1'b0)
        $display("FAIL [Path 2]");
```

```
    else
        $display("PASS [Path 2]");

    // Store values for Path 3 checking
    prev_P = P;
    prev_CARRYOUT = CARRYOUT;

    // Test case 3: Verify DSP Path 3
    A = 20; B = 10; C = 350; D = 25; CARRYIN = 1; BCIN = 18'h777; PCIN = 48'hF0F0F0;
    OPMODE = 8'b00001010;
    repeat(3) @(negedge CLK);
    if (BCOUT !== 18'ha || M !== 36'hc8 || P !== prev_P || PCOUT !== prev_P || CARRYOUT !==
prev_CARRYOUT)
        $display("FAIL [Path 3]");
    else
        $display("PASS [Path 3]");

    // Test case 4: Verify DSP Path 4
    A = 5; B = 6; C = 350; D = 25; CARRYIN = 1; BCIN = 18'hAAA; PCIN = 3000;
    OPMODE = 8'b10100111;
    repeat(3) @(negedge CLK);
    if (BCOUT !== 18'h6 || M !== 36'h1e || P !== 48'hfe6fffec0bb1 || PCOUT !==
48'hfe6fffec0bb1 || CARRYOUT !== 1'b1)
        $display("FAIL [Path 4]");
    else
        $display("PASS [Path 4]");

    // Reset all signals and finish
    RSTA = 1; RSTB = 1; RSTM = 1; RSTP = 1; RSTC = 1; RSTD = 1; RSTCARRYIN = 1; RSTOPMODE = 1;
    CEA = 0; CEB = 0; CEM = 0; CEP = 0; CEC = 0; CED = 0; CECARRYIN = 0; CEOPMODE = 0;
    A = 0; B = 0; C = 0; D = 0; CARRYIN = 0; BCIN = 0; PCIN = 0; OPMODE = 8'b00000000;
    repeat(10) @(negedge CLK);
    $stop;
end

//Test Monitor & Results
initial begin
    $monitor("A=%d, B=%d, C=%d, D=%d, CARRYIN=%d ,PCIN=%d , OPMODE=%b, P=%d,BCOUT=%d ,M=%d
,CARRYOUT=%d", A, B, C, D,CARRYIN ,PCIN , OPMODE, P,BCOUT ,M , CARRYOUT);
end
endmodule
```

3-Do File

```
vlib work
vlog BLK.v DSP.v DSP_tb.v
vsim -voptargs=+acc work.DSP48A1_tb
add wave *
run -all
#quit -sim
```

4-Waveform Snippet

Waveform Snippet



The waveform illustrates the sequential execution of the four directed test paths, driven by the OPMODE signal. The outputs (P, M, BCOUT, and CARRYOUT) correctly reflect the expected arithmetic operations (pre-addition/subtraction, multiplication, and accumulation) based on the applied data inputs.

Transcript Snippet

```
# A= 43981, B=  4660, C=     3735928559, D= 22136, CARRYIN=1 ,PCIN=       1193046 , OPMODE=11111111, P=              x,BCOUT=    x ,M=         x ,CARRYOUT=x
# A= 43981, B=  4660, C=     3735928559, D= 22136, CARRYIN=1 ,PCIN=       1193046 , OPMODE=11111111, P=              0,BCOUT=    0 ,M=         0 ,CARRYOUT=0
# PASS [Reset]: Outputs are zero
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=       5921370 , OPMODE=11011101, P=              0,BCOUT=    0 ,M=         0 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=       5921370 , OPMODE=11011101, P=              0,BCOUT=   10 ,M=         0 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=       5921370 , OPMODE=11011101, P=            350,BCOUT=   15 ,M=       200 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=       5921370 , OPMODE=11011101, P=            150,BCOUT=   15 ,M=       300 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=       5921370 , OPMODE=11011101, P=             50,BCOUT=   15 ,M=       300 ,CARRYOUT=0
# PASS [Path 1]
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      11259375 , OPMODE=00010000, P=             50,BCOUT=   15 ,M=       300 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      11259375 , OPMODE=00010000, P=              0,BCOUT=   35 ,M=       300 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      11259375 , OPMODE=00010000, P=              0,BCOUT=   35 ,M=       700 ,CARRYOUT=0
# PASS [Path 2]
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      15790320 , OPMODE=00001010, P=              0,BCOUT=   35 ,M=       700 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      15790320 , OPMODE=00001010, P=              0,BCOUT=   10 ,M=       700 ,CARRYOUT=0
# A=    20, B=    10, C=            350, D=    25, CARRYIN=1 ,PCIN=      15790320 , OPMODE=00001010, P=              0,BCOUT=   10 ,M=       200 ,CARRYOUT=0
# PASS [Path 3]
# A=     5, B=     6, C=            350, D=    25, CARRYIN=1 ,PCIN=          3000 , OPMODE=10100111, P=              0,BCOUT=   10 ,M=       200 ,CARRYOUT=0
# A=     5, B=     6, C=            350, D=    25, CARRYIN=1 ,PCIN=          3000 , OPMODE=10100111, P=              0,BCOUT=    6 ,M=       200 ,CARRYOUT=0
# A=     5, B=     6, C=            350, D=    25, CARRYIN=1 ,PCIN=          3000 , OPMODE=10100111, P=279756988484530,BCOUT=    6 ,M=        30 ,CARRYOUT=1
# A=     5, B=     6, C=            350, D=    25, CARRYIN=1 ,PCIN=          3000 , OPMODE=10100111, P=279756988484529,BCOUT=    6 ,M=        30 ,CARRYOUT=1
# PASS [Path 4]
# A=     0, B=     0, C=              0, D=     0, CARRYIN=0 ,PCIN=             0 , OPMODE=00000000, P=279756988484529,BCOUT=    6 ,M=        30 ,CARRYOUT=1
# A=     0, B=     0, C=              0, D=     0, CARRYIN=0 ,PCIN=             0 , OPMODE=00000000, P=              0,BCOUT=    0 ,M=         0 ,CARRYOUT=0
# ** Note: $stop    : DSP_tb.v(107)
#    Time: 48 ns  Iteration: 1  Instance: /DSP48A1_tb
```

5-Constraint File

```
## Clock signal
set_property -dict { PACKAGE_PIN W5   IOSTANDARD LVCMOS33 } [get_ports CLK]
create_clock -add -name sys_clk_pin -period 10.00 -waveform {0 5} [get_ports CLK]

## Configuration options, can be used for all designs
set_property CONFIG_VOLTAGE 3.3 [current_design]
set_property CFGBVS VCCO [current_design]

## SPI configuration mode options for QSPI boot, can be used for all designs
set_property BITSTREAM.GENERAL.COMPRESS TRUE [current_design]
set_property BITSTREAM.CONFIG.CONFIGRATE 33 [current_design]
set_property CONFIG_MODE SPIx4 [current_design]
```

6-Elaboration

Elaboration Schematic



Elaboration Message Tab



7-Synthesis

Synthesis Schematic



Synthesis Message Tab

Synthesis (58 warnings, 48 infos)

- [Common 17-349] Got license for feature 'Synthesis' and/or device 'xc7a200t'
- [Synth 8-2490] overwriting previous definition of module DSP48A1 [DSP.v:1]
- [Synth 8-6157] synthesizing module 'DSP48A1' [DSP.v:1] (5 more like this)
- [Synth 8-6155] done synthesizing module 'BLK' (1#1) [BLK.v:1] (5 more like this)
- [Synth 8-3331] design DSP48A1 has unconnected port CARRYIN (37 more like this)
- [Device 21-403] Loading part xc7a200tffg1156-3

Synthesis Utilization Report

Resource	Utilization	Available	Utilization %
LUT	230	134600	0.17
FF	143	269200	0.05
DSP	1	740	0.14
IO	327	500	65.40



Synthesis Timing Report

Setup

Worst Negative Slack (WNS):	5.168 ns
Total Negative Slack (TNS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	106

Hold

Worst Hold Slack (WHS):	0.182 ns
Total Hold Slack (THS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	106

Pulse Width

Worst Pulse Width Slack (WPWS):	4.500 ns
Total Pulse Width Negative Slack (TPWS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	145

All user specified timing constraints are met.

8-Implementation

Implementation Schematic



Implementation Message Tab

- Implementation (1 warning, 88 infos)
  - Design Initialization (7 infos)
    - [Netlist 29-17] Analyzing 207 Unisim elements for replacement
    - [Netlist 29-28] Unisim Transformation completed in 0 CPU seconds
    - [Project 1-479] Netlist was created with Vivado 2018.2
    - [Device 21-403] Loading part xc7a200tffg1156-3
    - [Project 1-570] Preparing netlist for logic optimization
    - [Opt 31-138] Pushed 0 inverter(s) to 0 load pin(s).
    - [Project 1-111] Unisim Transformation Summary:
      No Unisim elements were transformed.

Implementation Utilization Report

Summary

Resource	Utilization	Available	Utilization %
LUT	229	133800	0.17
FF	144	267600	0.05
DSP	1	740	0.14
IO	327	500	65.40



Implementation Timing Report

Design Timing Summary

Setup

Worst Negative Slack (WNS):	3.962 ns
Total Negative Slack (TNS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	107

Hold

Worst Hold Slack (WHS):	0.273 ns
Total Hold Slack (THS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	107

Pulse Width

Worst Pulse Width Slack (WPWS):	4.500 ns
Total Pulse Width Negative Slack (TPWS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	146

All user specified timing constraints are met.

FPGA Device



9-Linting Tool